HEWLETT hp PACKARD

HP-IB

MINICOMPUTER APPLICATIONS

AN 201-5

The HP-IB LINK: Control of Distributed HP-IB Devices

INTRODUCTION

With the increasing availability of HP-IB* compatible measurement and control devices engineers are finding they no longer must concern themselves with interconnection/interfacing problems. Instead, they can concentrate on the real problems of their applications. The HP-IB LINK permits single HP-IB instruments or clusters of HP-IB devices to be connected and controlled simultaneously by a remote HP1000 Computer System. This Application Note demonstrates the simplicity of connection and programming of a typical cluster of HP-IB devices via the LINK creating an automatic Voltage Controlled Oscillator (VCO) tester.

**The HP-IB (Hewlett-Packard Interface Bus) is Hewlett-Packard's implementation of IEEE Standard 488-1975 "Digital Interface for programmable instrumentation".*

FAST, EASY, ADAPTABLE CONSTRUCTION OF A NETWORK OF DISTRIBUTED HP-IB CLUSTERS.

The LINK is composed of a computer interface board, a shielded twisted pair cable of up to 4km long, and HP 3070A terminals (up to 56) connected along the cable. Connection between the cable and any terminal, or the system, is made through simple plugs and wall mounted sockets. Terminals may be connected and disconnected from the LINK without disturbance to others. This greatly simplifies the construction of a complete distributed system.

The cable is small and long enough to go anywhere, following any path to reach the HP-IB devices directly on their site. The system itself can also be located anywhere along this cable**. Shown below is a typical distributed network, and the detail of one HP-IB cluster is shown overleaf.

***Maximum distance between the system and the farthest HP-IB cluster is 2 km.*



A distributed HP-IB network through the HP-IB LINK.



A typical HP-IB cluster connected to the HP-IB LINK.

STRAIGHT FORWARD OPERATION

The HP 3070A terminals are used for the interfacing of HP-IB devices to the LINK. They also permits the operators to communicate with the remote system, and to intervene with the test process when required (start, stop, etc...).

The LINK uses a protocol based on a serial version of the HP-IB. Interface between the LINK and HP-IB is made through a communication module inside each terminal. Communication safety is ensured by the sophisticated LINK protocol and terminal circuitry which makes the network usable even in an electrically noisy environment.

The design of the HP 3070A Real Time Applications. Terminal is aimed at simplicity of use. A set of 15 user labelled prompting lights and a 16 digit numeric display conduct the operator during the different phases of an operation. A numeric keypad and a set of 10 application oriented special function keys allow him to easily input the required answers.

SIMPLE PROGRAMMING

Test or application engineers can develope their own programs with HP BASIC and a set of contributed HP 3070A utilities available from any HP sales office. Writing, debugging and running an application program need only be a matter of minutes. With the LINK, adding HP-IB instruments does not require any system shutdown or reconfiguration and does not disturb the existing users. A VCO Application is described in the following sections to illustrates the friendly power of the LINK.



A typical HP 3070A Label

A TYPICAL USE OF THE HP-IB LINK

The following example describes the use of an HP-IB based instrument cluster to automatically measure the transfer characteristic (frequency out as a function of voltage in) of a voltage controlled oscillator. Results are printed locally on a ticket for attachment to the tested unit. The measurement set up consists of a Digital to Analog Converter (HP 59303A) to generate 10 digitally controlled analog inputs to the VCO under test. Each DAC output voltage is measured with a DVM (HP 5300B / 5306A / 5312A) while each VCO output frequency is measured by a counter (HP 5300B / 5302A/ 5312A).

Control signals to the HP-IB instruments, and the measurement results are connected to the system through the HP 3070A terminal. Results are printed on the local printer HP 5150A. Using a typical HP 1000 System configuration, with little other activity occuring, the measurement part of the test written in Fortran was completed within 6 to 15 seconds (9 to 20 seconds for same program written in Basic) with the HP 3070A address varying from 1 to 10. Add 7 to 10 seconds (9 to 11 in Basic) for printing the results on the local strip printer. The variation in speed obtained in systems with multiple clusters will depend mostly on total system activity and the amount of swapping occuring.



VCO Test Set-up

PROGRAM FLOW CHART

Manualy entered Data



Automatic test



PROGRAM LISTINGS

The statements shown highlighted in the listing are the HP 3070A utilities to control the terminal itself and to accomplish all HP-IB control and data input/output functions through the terminal.

```
10  REM          *************************
20  REM          *                       *
30  REM          *     V.C.O. TEST       *
40  REM          *                       *
50  REM          *    MHB/GO 3/21/77     *
60  REM          *      GRENOBLE         *
70  REM          *************************
80  REM
90  REM
100 DIM V$[20],A$[20],C$[20],P$[5],F$[20],V[11],F[11],T$[20],Z$[2]
110 PRINT " ENTER 3070'S LU NUMBER : ";
120 INPUT L
130 REM
140 REM ========== CLEAR THE COMPLETE SET-UP ==========
150 LET C=CLR(L)
160 REM
170 REM ========== ENABLE ALL SRQ ==========
180 LET S1=SRQ(L,11)
190 REM
200 REM                  ***************************
210 REM *************** START OF MANUAL KEYING ***************
220 REM                  ***************************
230 REM
240 REM ========== LIGHT "READY FOR NEXT TEST " ON ==========
250 LET L1=LIT(L,1,0,0)
260 REM
270 REM ========== WAIT FOR OPERATOR INTERVENTION ==========
280 CALL RDKBD(L,B$,S2,R)
290 IF R=1 THEN 150
300 IF S2=2 THEN 380
310 IF S2=10 THEN 1780
320 REM
330 REM ===== LIGHT "ERROR" ON =====
340 LET L1=LIT(L,5,0,0)
350 GOTO 280
360 REM
370 REM ========== LIGHT "INSTRUMENT SERIAL #" ON ==========
380 LET L1=LIT(L,-1,-5,6)
390 REM
400 REM ========== GET OPERATOR'S ANSWER ==========
410 LET C$="      "
420 CALL RDKBD(L,C$,S2,R)
430 IF R=1 THEN 150
440 IF LEN(C$) <= 6 THEN 520
450 REM
460 REM ===== LIGHT "ERROR" ON =====
470 LET L1=LIT(L,5,0,0)
480 LET C$="      "
490 GOTO 420
500 REM
510 REM ========== LIGHT "OPTION # ?" ON ==========
520 LET L1=LIT(L,-5,-6,7)
530 LET B$[1,6]=C$[1,6]
540 REM
550 REM ========== GET OPERATOR'S ANSWER ==========
560 LET C$=""
570 CALL RDKBD(L,C$,S2,R)
580 IF R=1 THEN 150
590 IF LEN(C$) <= 3 THEN 660
600 LET L1=LIT(L,5,0,0)
610 LET C$="    "
620 GOTO 570
630 REM
640 REM ========== LIGHT "TEST IN PROGRESS" ON ==========
650 REM ========== AND DISPLAY ERAZED ==========
660 LET L1=LIT(L,-7,-5,8)
670 CALL TWRIT(L,0)
680 PRINT #L;"    "
690 LET B$[7,9]=C$[1,3]
700 REM
710 REM                  *****************************
720 REM                  *   END OF MANUAL KEYING    *
730 REM ***************                               ***************
740 REM                  *  START OF AUTOMATIC TEST  *
750 REM                  *****************************
760 REM
770 REM ========== SET REMOTE ENABLE ==========
780 LET R=REN(L,1)
790 REM
800 REM ===== INITIALIZE THE D.A.C. AND DISABLE LOCAL PUSHBUTTON ====
810 LET C=CMD$T(L,"_?% ","E0")
820 LET P$="      "
830 LET K=0
840 REM
850 REM ========== LOOP TO PROGRAM THE DIFFERENT OUTPUT VOLTAGES ====
860 FOR V=0 TO -100 STEP -10
870 CALL DCODE(V,P$,"(F4.0)")
880 CALL TWRIT(L,4)
890 PRINT #L;P$
900 REM ===== CHECK IF TEST ABORTED BY OPERATOR =====
910 LET B=RSTAT(L,29)
920 IF B=64 THEN 1660
930 REM
940 REM ===== LOOP TO ALLOW D.A.C. TO STABILIZE TO WITHIN 10 MV. ====
950 FOR M=1 TO 10
960 REM
970 REM == MONITOR D.A.C. OUTPUT WITH DVM ==
980 CALL TWRI(L,3)
990 PRINT #L;"I"
1000 CALL TREAD(L,3)
1010 READ #L;V$
1020 CALL DCODE(V$[5,15],A,"(E10.4)")
1030 IF ABS(ABS(10*V)-ABS(1000*A)) <= 10 THEN 1100
1040 NEXT M
1050 REM
1060 REM ===== CANNOT STABILIZE THE D.A.C. =====
1070 REM
1080 REM == LIGHTS "ERROR" "STOP" AND "D.A.C. NOT STAB'D" ON ==
1090 LET L1=LIT(L,5,10,15)
1100 REM
1110 REM == WAIT FOR OPERATOR INTERVENTION ==
1120 CALL RDKBD(L,C$,S2,R)
1130 IF R=1 THEN 150
1140 IF S2=10 THEN 1750
1150 GOTO 1120
1160 REM
1170 REM ========== GROUP TRIGGER THE DVM AND THE COUNTER ==========
1180 REM ========== TO MEASURE D.A.C. AND V.C.O. OUTPUTS ==========
1190 LET Z$="AA"
1200 CALL CHR$(A,Z$)
1210 LET C=CMD$T(L,"_?#1",Z$)
1220 CALL TREAD(L,3)
1230 READ #L;V$
1240 CALL TREAD(L,1)
1250 READ #L;F$
1260 LET K=K+1
1270 CALL DCODE(V$[5,15],V[K],"(E10.4)")
1280 LET V[K]=1000*V[K]
1290 CALL DCODE(F$[4,15],F[K],"(E11.4)")
1300 LET F[K]=F[K]/1000
1310 NEXT V
1320 REM
1330 REM ========== PRINT RESULTS ON LOCAL STRIP PRINTER ==========
1340 LET T$="     END OF TEST     "
1350 CALL TWRIT(L,6)
1360 PRINT #L;""
1370 PRINT #L;T$
1380 PRINT #L;""
1390 LET V$="     "
1400 LET F$="          "
1410 FOR K=1 TO 11
1420 CALL DCODE(V[K],V$,"(F4.0)")
1430 CALL DCODE(F[K],F$,"(F9.3)")
1440 LET T$[1,2]="  "
1450 LET T$[3,7]=V$[1,5]
1460 LET T$[8,11]="  :  "
1470 LET T$[12,20]=F$[1,9]
1480 PRINT #L;T$
1490 NEXT K
1500 PRINT #L;""
1510 PRINT #L;"    -V     :     KHZ    "
1520 PRINT #L;" VOLTAGE : FREQUENCY"
1530 PRINT #L;""
1540 LET C$=" OPTION  :           "
1550 LET C$[12,14]=H$[7,9]
1560 PRINT #L;C$
1570 LET C$=" NUMBER  :           "
1580 LET C$[12,17]=B$[1,6]
1590 PRINT #L;C$
1600 PRINT #L;""
1610 PRINT #L;"   == VCO TEST ==   "
1620 FOR Y=1 TO 6
1630 PRINT #L;""
1640 NEXT Y
1650 REM
1660 REM ========== RESET D.A.C. ==========
1670 CALL TWRIT(L,4)
1680 PRINT #L;"0"
1690 REM
1700 REM                  ***************************
1710 REM *************** END OF AUTOMATIC TEST ***************
1720 REM                  ***************************
1730 REM
1740 GOTO 150
1750 LET L1=LIT(L,-16,0,0)
1760 REM
1770 REM ========== RESET D.A.C. ==========
1780 CALL TWRIT(L,4)
1790 PRINT #L;"0"
1800 REM
1810 REM ========== CLEAR CLUSTER ==========
1820 LET C=CLR(L)
1830 END
```

HEWLETT hp PACKARD

1501 Page Mill Road, Palo Alto, California 94304

Printed in U.S.A. - Sept 1977 - 5953-0114 (63)